WARNING: notes below are useless unless you actually type them on the

screen and see how it works. Copy-and-paste is also possible, but at the

beginning I recommend typing - 99% of errors comes from typing errors, and it is better to get reed of them from the start. Below mathematica commands are highlighted in bold (just as the section headings). I often give them together with In[...]:= , the way you see it on the screen. Check that you indeed get the same Out[...]. Note: symbols (rather than script) for integrals etc. are also available, e.g from pallets


INTRODUCTION TO MATHEMATICA

TOPICS:

0. GETTING STARTED


entering a command


help


useful type- and space saving commands


saving files


brackets


equality signs

I. NUMBERS:


exact


approximate


complex


numnbers with dimensions

II. SYMBOLIC MATH


sums and series; integration


algebra


trigonometry


Taylor expansions and limits

III  GRAPHICS

2D


3D
TOPIC 0. GETTING STARTED.

1. Just try to open a Mathematica notebook. 
2. Entering a number or command: Shift_enter 

All functions have square brackets, [], get used to it!

Try to use Mathematica as a simple calculator - a useful warm-up. 

3. Help. 

 1) if you know the exact command , but want to refresh what argument it requires, use ?

E.g. 

In[9]:= ?Sin

Sin[z] gives the sine of z.  

 2) if you approximately know the spelling, use ? with * for the unknown part, e.g.

In[10]:= ?*Plot*

gives all commands which have Plot in them:

ListDensityPlot  Plot             PlotLabel        PlotStyle

etc.

3) if you have no idea - use Help button 

4. Frequent type- and space-saving commands:


1) % uses the last output as input. E.g.

In[12]:= Sin[1.]

Out[12]= 0.841471

In[13]:= ArcSin[%]

Out[13]= 1. 

       Similarly, %% uses the one before last output, etc. Or, %12 will use the output from the 12th line

In[14]:= %12
Out[14]= 0.841471 


2) space - can be used instead of * for multiplication:

In[1]:= 2 2

Out[1]= 4

    Do not need space if number in front of a symbol:

In[2]:= 2x

Out[2]= 2 x

(useful to keep  close to hand-written formulas)

but will keep a symbol if number without a space

In[3]:= x2

Out[3]= x2  

don't forget space if you intend multiplication:

In[4]:= x 2

Out[4]= 2 x


3) ; will not produce an output on the screen (but can work with it further!) 

In[5]:= Factorial[1000];
In[6]:= N[%]

                            2567

Out[6]= 4.023872600770938 10       

(main typesaving - defining your own functions, etc. - will study later).

5. Saving your work. 

There are  two ways:

1) Save["filename", symbol] appends definitions associated with the specified symbol to a file.

if symbol includes previous definitions, will save everything which is

required! "filename" usually includes .m at the end (for convenience), but you can be creative. Graphics should not be saved this way, but you can save the last command used to generate it, and then recreate the picture upon restarting Mathematica. Files are in plain text and relatively small.

Example: 

In[1]:= fig:=Plot[Sin[x]/x, {x,-8,8}] (we defined a plot function, fig)

In[2]:= Save["figSinc.m", fig] (saved this function in a file figSinc.m)

In[3]:= !!figSinc.m
(this shows the contents of the file)

fig := Plot[Sin[x]/x, {x, -8, 8}]

Now, if you start a new Mathematica session, you can type 


<< figSinc.m 

and you will have all saved definitions. Command fig will plot your picture. [Note: sometimes will save in another directory –depends on Windows; then need to show path for “<<” and !! may not  work to view an external file].
2) you save as a notebook, with all graphics you created (and all the junk). Saved files are BIG, and can quickly overflow your directory if caution is not used. Use sparingly, and only for work you feel you really need.
6. Brackets

( ) – as in regular math

[ ] – argument of ANY function: Sin, Plot, etc. Can be empty, e.g. RandomInteger[]
{ } – lists, e.g. {1,2,3}

7. Equal signs

= - immediate assignment

:= - delayed. Try 
r:=RandomInteger[]; Table[r,{5}];

and compare to 
Clear[r]; r=RandomInteger[];Table[r,{5}];
== - logical, e.g. Solve[x^2==4,x] or
Cos[p]^2+Sin[p]^2==1//Simplify 

True
TOPIC I. NUMBERS:

Integer: remarkable things,

Factorial[1000], Divisors[10^10+27], Prime[1000000], etc. Try it!

1. exact numbers - 1/2, 10^-10, Pi, E, Sqrt[2], etc.


e.g. 2^2-4=0 (exactly!)



Sin[Pi/2] =1



Log[E] = 1

(Note, all functions and constants provided by Mathematica start with a capital; your own can be any, usually with lower case to distinguish)

2. approximate numbers - 2. , 10.^-10,  pi= N[Pi,15], e=N[E,7], etc.

Why use approximate?

1) With exact data will not give an answer, unless "knows" it.

In[51]:= Sqrt[Pi]

Out[51]= Sqrt[Pi]  

but

In[52]:= N[Sqrt[Pi]]

Out[52]= 1.77245

2) An exact result can be ridiculously long, e.g. Factorial[1000], but

In[53]:= Factorial[1000]//N

                             2567

Out[53]= 4.023872600770938 10    

3) Calculation with approximate numbers are much faster

Caution: can get errors with approximate numbers when adding very big and very small together 

3. Advanced: Complex numbers:


I represents the imaginary unit Sqrt[-1]

In[59]:= Solve[X^2+1==0,X]

Out[59]= {{X -> -I}, {X -> I}}

In[60]:= z=1+I 2

Out[60]= 1 + 2 I

In[61]:= Re[z]

Out[61]= 1

In[62]:= Abs[z]

Out[62]= Sqrt[5]

In[64]:= Arg[z]

Out[64]= ArcTan[2]

Caution: gives one branch for branching functions (Sqrt, Log, etc.):

In[68]:= Log[2.+3 I]

Out[68]= 1.28247 + 0.982794 I

In[70]:= Exp[%]

Out[70]= 2. + 3. I

In[71]:= %68 +2 Pi I

Out[71]= 1.28247 + 7.26598 I

In[72]:= Exp[%]

Out[72]= 2. + 3. I             (i.e. the same answer, although we added 2Pi*I to the logarithm) 

4. NUMBERS with DIMENSIONS:

Clear[m,s,n,kg]; cm = m/100.; mm = m/1000.; nm = m/1000000000.; A = 10.^-10m;g=9.8m/s^2; gram=10.^-3kg;
(these are the new units defined by us; n will be used for Newtons, not N, since N is already used in Mathematica) 

lengths = {cm, mm, nm, A} 

Note: lengths is a List, and can be treated as a single object

(will study later). 

Now let us construct our 1st formula - the 2nd law:

Clear[M,a];force=M a; (*note – another M*)
we can now replace M and a by numbers using /. - the

replacement operation

force/.{M->1gram, a->1 mm/s^2}

gives

… kg m/s^2
If you like it, Save["newt.m", lengths, force] .

Caution: since standard notations for dimensions are very short, make sure the same symbol is not used for something else, i.e. m ("meters") should remain "free". Use


Clear[m] 

if you are not sure.

-----------------------------------------------------------------

HW1: construct some realistic energy-dependent formula which would

"understand" Joules , watts and hp
-------------------------------------------------------------------

Note: Mathematica does not know that m, s. etc are > 0. Thus, use

PowerExpand to simplify dimensions:

In[59]:= g=9.8 m/s^2 ; 

In[60]:= t=Sqrt[2 h/g] ;

In[61]:= t/.h->10 m

                       2

Out[61]= 1.42857 Sqrt[s ]

In[62]:= PowerExpand[%]

Out[62]= 1.42857 s         
TOPIC 2. SYMBOLIC MATH:

1. Sums and series; derivatives and integration

In[74]:= Sum[i^2, {i,1,n}]

         n (1 + n) (1 + 2 n)

Out[74]= -------------------

                  6           

In[63]:= Sum[i^-2, {i, 1, Infinity}]

           2

         Pi

Out[63]= ---

          6         

Derivatives and integration:

In[75]:= D[x^n,x]
            -1 + n

Out[75]= n x

In[76]:= D[%,x]

                     -2 + n

Out[76]= (-1 + n) n x

In[77]:= Integrate[%,x]

            -1 + n

Out[77]= n x

Use help:

In[91]:= ?Integrate

Integrate[f, x] gives the indefinite integral of f with respect to x. Integrate[f, {x, xmin, xmax}] gives the definite integral of f with respect to x from xmin to xmax. 

Assumptions:

In[1]:= Integrate[Exp[-a x], {x,0,Infinity}]

                      1             -(a x)

Out[1]= If[Re[a] > 0, -, Integrate[E      , {x, 0, Infinity}]]

                      a

(the program does not “know” if a>0 and is unsure if the integral converges.

Can use Assumptions -> a>0 )

In[2]:= Integrate[Exp[-a x], {x,0,Infinity}, Assumptions -> a>0]

        1

Out[2]= -

        a                                                                    

2. Algebra:

In[93]:= Expand[(1+x)^4]

                      2      3    4

Out[93]= 1 + 4 x + 6 x  + 4 x  + x

In[94]:= Factor[%]

                4

Out[94]= (1 + x)

In[95]:= Simplify[%93]

                4

Out[95]= (1 + x)

In[100]:= Collect[%94,x]

                       2      3    4

Out[100]= 1 + 4 x + 6 x  + 4 x  + x

3. Trigonometry:

In[5]:= TrigExpand[Cos[5 x]]

              5            3       2                  4

Out[5]= Cos[x]  - 10 Cos[x]  Sin[x]  + 5 Cos[x] Sin[x]

In[7]:= TrigReduce[Cos[x]^5]

        10 Cos[x] + 5 Cos[3 x] + Cos[5 x]

Out[7]= ---------------------------------

                       16                   

Can use Simplify to verify identities:

In[8]:= Simplify[%-Cos[x]^5==0]

Out[8]= True

(Note: == for logical equal)

4. (Advanced)  Symbolic computations with  complex numbers 

Connection with exponential notations:

In[8]:=  ExpToTrig[Exp[I x]]

Out[8]= Cos[x] + I Sin[x] 

or

In[9]:= TrigToExp[Cos[x]+I Sin[x]]

           I x

Out[9]= E                    

Simplification of expressions with complex numbers. Difficulty: Mathematica does not know which symbol is complex and which is real. There are 2 options:

1) use Simplify with assumptions, e.g.

In[13]:= Simplify[Conjugate[a+I b], Assumptions->{Element[a, Reals],Element[b,Reals]}]
Out[13]= a - I b

(but Simplify is not very well suited for complex numbers, so will not work with Arg[I b])

2) use ComplexExpand with the option TargetFunctions -> {Re, Im}, e.g.

In[16]:= ComplexExpand[Arg[I b], TargetFunctions -> {Re, Im}]

Out[16]= ArcTan[0, b] 

Or

F=1/(a+I b) ;  ComplexExpand[Im[F]]

-(b/(a2+b2))

5. Advanced. Power series and limits:

In[118]:= Series[Exp[a x], {x, 0, 5}]

                     2  2    3  3    4  4    5  5

                    a  x    a  x    a  x    a  x        6

Out[118]= 1 + a x + ----- + ----- + ----- + ----- + O[x]

                      2       6      24      120

To make a polynomial by truncating a series:

In[119]:= Normal[%]

                     2  2    3  3    4  4    5  5

                    a  x    a  x    a  x    a  x        

Out[119]= 1 + a x + ----- + ----- + ----- + ----- 

                      2       6      24      120

limit can exist while a power series does not:

In[123]:= Limit[(1+x/n)^n, n->Infinity]

           x

Out[123]= E                                                                 

Directed limits for cusped functions:

Plot[Sign[x],{x,-2,2}]

(this is just to see on the screen; we shall talk about plotting

separately)

In[125]:= Limit[Sign[x], x->0]

Out[125]= 1                  

(from above by default, direction -1)

In[127]:= Limit[Sign[x], x->0, Direction -> 1]

Out[127]= -1   

Singular functions:

Series[1/Sin[z],{z,0,3}]

1/z+z/6+(7 z3)/360+O[z]4
Advanced. The “Do” loop

(in class)

Do[Print[𝑛^2],{𝑛,4}]
1

4

9

16

(GO TO IntroFun.nb)
------------------------------------------------------------
(*              

III. GRAPHICS:

TOPICS:

Two-dimensional 

Three-dimensional and color

Two-dimensional:

Main functions: Plot, PlotRange, Show

Plot[f, {x, xmin, xmax}] generates a plot of f as a function of x from xmin to xmax. Plot[{f1, f2, ... }, {x, xmin, xmax}] plots several functions fi.

PlotRange is an option for graphics functions that specifies what points to include in a plot.

Show[graphics, options] displays two- and three-dimensional graphics, using the options specified. Show[g1, g2, ... ] shows several plots combined.

Use with Evaluate for complicated functions
Dashing of lines 
Dashing[{r1, r2, ... }] is a two-dimensional graphics directive which specifies that lines which follow are to be drawn dashed, with successive segments of lengths r1, r2, ...  (repeated cyclically). The ri is given as a fraction of the total width of the graph.

In[12]:= Clear[plo]     

In[13]:= plo[n_]:=Plot[Sin[n x]/x, {x,-2,2}, PlotRange -> {-1,2}, PlotStyle -> Dashing[{0.01*n, 0.02}]]

In[14]:= plo[1]

Out[14]= -Graphics- (*HW: try it!*)

In[15]:= sho:=Show[Table[plo[n], {n,1,3}]]

(*Note: to plot Tables of more complex functions, use Plot[Release[Table[…]]] *)

In[16]:= sho

Out[16]= -Graphics-

[image: image1.png]



AxesLabel -> {“x”, “y”} will label each axes.

Greek letters: \[Alpha], etc. (in Notebook can copy from pallet)

(*HW: label x,y with alpha and beta*)

Plotting discrete data points:

ListPlot[{y1, y2, ... }] plots a list of values. The x coordinates for each point are taken to be 1, 2, ... . ListPlot[{{x1, y1}, {x2, y2}, ... }] plots a list of values with specified x and y coordinates.

In[21]:= list=Table[Sin[i/100.]+.1*RandomReal[], {i,400}];

In[22]:= ListPlot[list]

Out[22]= -Graphics-

[image: image2.png]



Main extra options:e.g.,  PlotStyle -> PointSize[0.02],

Or Joined->True

(*HW: try this *)

Parametric plot:

ParametricPlot[{fx, fy}, {t, tmin, tmax}] produces a parametric plot with x and y coordinates fx and fy generated as a function of t.   ParametricPlot[{{fx, fy}, {gx, gy}, ... }, {t, tmin, tmax}] plots several parametric curves.

 x[phi_]:=Cos[phi]; y[phi_]:=Sin[phi];

 ParametricPlot[{x[phi],y[phi]}, {phi, 0, 2Pi}]

 -Graphics- (not a circle)

(*by default, in some versions AspectRatio ->GoldenRatio*)

(*HW: try this!*Show[%, AspectRatio -> Automatic]

 -Graphics-

(*HW: plot a spiral*)

Text: 

mytext=Graphics[Text[“something”, {1,1}]]; 

(*then use Show[…, mytext] with other graphics, e.g., -x*)

[image: image4.png]“omething




Labels as parameters:

plo[n_] := Plot[x^n, {x, 0, 1}, PlotLabel ->n] (*no quotes now!!!*)

s

0l

03

0z

01




(*if we like what we see and want to create, e.g. a postscript file, t.ps *)

disp:=Export["t.ps", #, "EPS"]&
(*now disp[plo[3]] will create t.ps, as a GOOD postscript*)

 Other things: Graphics Primitives:

Line[{pt1, pt2, ... }] is a graphics primitive which represents a line joining a sequence of points. Arrow - similar
Point[coords] is a graphics primitive that represents a point.

Circle[{x, y}, r] is a two-dimensional graphics primitive that represents a circle of radius r centered at the point x, y. Circle[{x, y}, {rx, ry}] yields an ellipse with semi-axes rx and ry. Circle[{x, y}, r, {theta1,theta2}] represents a circular arc.

Polygon, etc. – use with Graphics, similar to Arrow

(*HW: give graphic examples*).

(* 

Advanced: 3-Dimensional graphics and colors
*)

"Plot3D[f, {x, xmin, xmax}, {y, ymin, ymax}] generates a three-dimensional plot of f as a function of x and y
"

Color:

"RGBColor[red, green, blue] is a graphics directive which specifies that graphical objects which follow are to be displayed, if possible, in the color given." ( “knows” main colors, Red,Blue,etc. )

 (* in 2-D *)

Plot[{Sin[x], Cos[2, x]}, {x, 0, 5}, PlotStyle ->{Red, Green}]

(*HW: try this*)

(* Color can be a function: *)

Plot3D[{Sin[x]/x,RGBColor[Abs[Sin[x]/x],1-Abs[Sin[x]/x],x-Floor[x]]},{x,-8,8},{y,-5,5}]

[image: image6.png]



(* ParametricPlot3D:

"ParametricPlot3D[{fx, fy, fz}, {t, tmin, tmax}] produces a three-dimensional space curve parametrized by a variable t which runs from tmin to tmax. \

ParametricPlot3D[{fx, fy, fz}, {t, tmin, tmax}, {u, umin, umax}] produces a three-dimensional surface parametrized by t and u. 
" *)

(* before looking at any of the examples below, try to do them on your own! *)

(*Plot a cone:*)

cone := ParametricPlot3D[{Cos[u]*v, Sin[u]*v, 1 - v}, {u, 0, 2*Pi}, {v, 0, 1}, PlotPoints -> {60, 12}]

[image: image7.png]



(*Note: by default, few points are used for 3D; PlotPoints option with a larger number makes it better (and slower)*)

(*Plot a sphere:*) 

sphere := ParametricPlot3D[{Cos[u]*Sin[v], Sin[u]*Sin[v], Cos[v]}, 

   {u, 0, 2*Pi}, {v, 0, Pi}, PlotPoints -> {60, 12}]

[image: image8.png]



Density plot:

"DensityPlot[f, {x, xmin, xmax}, {y, ymin, ymax}] makes a density plot of f as a function of x and y"

DensityPlot[Sin[x/y],{x,-1,1},{y,-1,1},ColorFunction ->Hue]

(* Black and white:

Optional: try to plot a chessboard : *)

f[x_,y_]:= If[EvenQ[Floor[x+y]],-1,1]

DensityPlot[f[x,y],{x,1,8},{y,1,8}, PlotPoints ->8]

(* Contour Plot *)
"ContourPlot[f, {x, xmin, xmax}, {y, ymin, ymax}] generates a contour plot of f as a function of x and y."

ContourPlot[x^2+y^2,{x,-1,1}, {y, -1,1}]

(*HW: plot this*)

e[x_,p_]:=u[x]+p^2/2

(*Example: Physical pendulum*)

u[x_]:=1-Cos[x]

ContourPlot[e[x,p], {x, -2 Pi, 2 Pi}, 


{p, -2, 2}]

(* too few default points for a nice plot *)

(*HW(optional) try*)

ContourPlot[e[x,p], {x, -2 Pi, 2 Pi}, 


{p, -2, 2}, PlotPoints->100]

(* what if want color ? *)

ContourPlot[e[x,p], {x, -2 Pi, 2 Pi}, 


{p, -2, 2}, PlotPoints ->100, ColorFunction ->Hue]

[image: image10.png]



(* what if do not need color? *)

ContourPlot[e[x,p], {x, -4 Pi, 4 Pi}, 


{p, -2, 2}, PlotPoints ->100, ContourShading 


-> False, AspectRatio ->.3]

(* AspectRatio for a better looking  (elongated) graph *)

[image: image11.png]





(*IMPLICIT PLOT – from ContourPlot; *)
ContourPlot[x^2+y^21,{x,-1,1},{y,-1,1}] (*circle*)

ContourPlot[{(x^2+y^2)^2==x^2-y^2,(x^2+y^2)^22 x y},{x,-1,1},{y,-1,1},ContourStyle{GrayLevel[0],Dashing[{.03}]}] (*2plots with different styles – see output below. ContourPlot3D – similar*)

